

Применение формул массивов

В ЭТОЙ ГЛАВЕ ...

- Одноячеечные формулы массивов
- Многоячеечные формулы массивов

В предыдущей главе были введены понятия массива и формулы массива, а также даны соответствующие примеры. Настоящая глава посвящена формулам массивов и содержит полезные примеры, которые продемонстрируют их эффективность.

Эти примеры вы можете непосредственно использовать в работе, только в ряде случаев следует изменить имена диапазонов или ссылок. Во многие примеры легко внести изменения, чтобы они работали именно так, как нужно для решения практических задач.



На сайте

Все примеры этой главы содержатся в русифицированном файле `Chapter15_RUS.xlsx` (названия листов соответствуют номерам рисунков), а также в нескольких авторских файлах, о которых будет сказано дополнительно.

Одноячеечные формулы массивов

Как описывалось в предыдущей главе, формулу массива можно вводить в одну ячейку (а не в диапазон ячеек). Такие формулы обрабатывают массивы, хранящиеся либо в диапазоне ячеек, либо в памяти Excel. В этом разделе представлены некоторые дополнительные примеры одноячеечных формул массива.



О ПРИМЕРАХ ЭТОЙ ГЛАВЫ

Данная глава содержит ряд примеров формул массива. Помните, что формулы массива вводятся с помощью комбинации клавиш `<Ctrl+Shift+Enter>`, а не просто клавиши `<Enter>`. После нажатия этой комбинации клавиш Excel вставит введенную вами формулу в фигурные скобки. Приведенные в тексте главы формулы массива заключены в фигурные скобки только для наглядности, вам не придется вводить их в строке формул.



На сайте

Все примеры этого раздела в авторском варианте содержатся в файле `single-cell array formulas.xlsx`.

Суммирование в диапазоне, содержащем ошибки

Многие пользователи Excel знают, что функция СУММ не работает, если просуммировать диапазон ячеек, содержащих сообщения об ошибках (например, #ДЕЛ/0! или #Н/Д). На рис. 15.1 показан такой случай. Формула СУММ в ячейке С11 возвращает сообщение об ошибке, так как в диапазоне ячеек С4:С10, которые она складывает, содержатся сообщения об ошибках.

	C11						
1	Суммирование диапазона со значениями ошибок						
2							
3	Всего	Число	Удельно				
4	80	10	8				
5	120	6	20				
6	144	12	12				
7			#ДЕЛ/0!				
8			#ДЕЛ/0!				
9	100	20	5				
10	50	5	10				
11	ВСЕГО:		#ДЕЛ/0!				
12							
13			55,000	<-- СУММ, исключая ошибки			
14							

Рис. 15.1. Формула массива может складывать значения в диапазоне, содержащем ошибки

Следующая формула массива возвращает сумму значений, даже если входной диапазон содержит сообщения об ошибке:

```
{=СУММ(ЕСЛИОШИБКА(С4:С10;"") )}
```

Эта формула создает новый массив, включающий значения диапазона, но без сообщений об ошибке. Функция ЕСЛИ заменяет сообщения об ошибке пустыми значениями. Затем функция СУММ складывает элементы “отфильтрованного” таким образом массива. Аналогичную методику можно использовать и для других функций, например МИН и МАКС.

В данном примере функция СУММ обрабатывает следующий внутренний массив, хранящийся в памяти:

```
{8;20;12;"";"";5;10}
```



Примечание

Возникают случаи, когда вместо ЕСЛИОШИБКА необходимо использовать другие функции. Дело в том, что ЕСЛИОШИБКА возвращает значение ИСТИНА при любых значениях ошибки (#Н/Д, #ЗНАЧ!, #ССЫЛКА!, #ДЕЛ/0!, #ЧИСЛО!, #ИМЯ? или #ПУСТО!). Функция ЕОШ возвращает значение ИСТИНА при любых значениях ошибки, кроме #Н/Д, а функция ЕНД — только при значениях ошибки #Н/Д.



Совет

Функция АГРЕГАТ предоставляет еще один способ суммирования диапазона, содержащего ошибочные значения:

```
=АГРЕГАТ(9,2,С4:С10)
```

Первый аргумент содержит значение 9 — код функции СУММ. Второй аргумент приказывает функции АГРЕГАТ игнорировать ошибки.

Подсчет количества ошибок в диапазоне

Представленная ниже формула массива подобна формуле предыдущего примера, за исключением того, что она возвращает количество ошибок в диапазоне Данные.

```
{=СУММ(ЕСЛИ(ЕОШИБКА(Данные);1;0))}
```

Эта формула создает массив, состоящий из единиц (если соответствующая ячейка содержит сообщение об ошибке) и нулей (если в ячейке отсутствует сообщение об ошибке).

Можно немного упростить формулу, удалив третий аргумент функции ЕСЛИ. В результате, если условие не выполняется (ячейка не содержит сообщение об ошибке), функция ЕСЛИ будет возвращать значение ЛОЖЬ. Усовершенствованная формула возвращает такой же результат, что и предыдущая, но в ней нет третьего аргумента функции ЕСЛИ.

```
{=СУММ(ЕСЛИ(ЕОШИБКА(Данные);1))}
```

В действительности можно еще больше упростить эту формулу:

```
{=СУММ(ЕОШИБКА(Данные)*1)}
```

В этой версии формулы используются такие равенства:

ИСТИНА*1=1

и

ЛОЖЬ*1=0

Суммирование *n* наибольших значений в диапазоне

Следующая формула возвращает значение суммы десяти наибольших значений в диапазоне Данные:

```
{=СУММ(НАИБОЛЬШИЙ(Данные;СТРОКА(ДВССЫЛ("1:10"))))}
```

Функция НАИБОЛЬШИЙ выполняется десять раз, причем каждый раз с другим вторым аргументом (1, 2, 3... 10). Результаты этих вычислений хранятся в новом массиве, который служит аргументом для функции суммирования.

Чтобы просуммировать другое количество значений, необходимо заменить число 10 в аргументе функции ДВССЫЛ на другое значение.

Если количество суммируемых ячеек находится в ячейке C17, используйте оператор конкатенации для создания адреса диапазона для функции ДВССЫЛ.

```
{=СУММ(НАИБОЛЬШИЙ(Данные;СТРОКА(ДВССЫЛ("1:"&C17))))}
```

Для суммирования *n* наименьших значений в диапазоне следует заменить функцию НАИБОЛЬШИЙ функцией НАИМЕНЬШИЙ.

Вычисление среднего без учета нулевых значений

На рис. 15.2 показан рабочий лист, на котором вычисляется средний объем продаж. В ячейку B14 помещена следующая формула:

```
=СРЗНАЧ(Данные)
```

	A	B	C	D	E	F	G
1	Исключение нуля из процедуры усреднения						
2							
3							
4	Менеджер	Продажи					
5	Иванов	23 991					
6	Петров	15 092					
7	Сидоров	0					
8	Кузнецов	11 893					
9	Гончаров	32 116					
10	Козлов	29 089					
11	Гаррисон	0					
12	Перельман	33 211					
13							
14		18 174	<-- Усреднение с учетом нулей				
15		24 232	<-- Усреднение без учета нулей (формула массива)				
16		24 232	<-- Усреднение без учета нулей (обычный способ)				
17							

Рис. 15.2. Вычисление среднего значения, при котором исключаются ячейки, содержащие нуль

Эта формула вычисляет среднее значение именованного диапазона Данные. Поскольку два продавца отсутствовали на работе некоторое время, в результате представленная выше формула неточно отображает средний уровень продаж торговых представителей.



Функция СРЗНАЧ не учитывает пустые ячейки, зато принимает в расчет ячейки, содержащие нулевые значения.

Примечание

Следующая формула массива возвращает среднее значение в диапазоне и при этом исключает ячейки, содержащие нулевые значения:

```
{=СРЗНАЧ(ЕСЛИ(Данные<>0;Данные))}
```

Эта формула создает новый массив, состоящий исключительно из ненулевых значений. Функция СРЗНАЧ принимает этот массив в качестве своего аргумента. Тот же результат можно получить при использовании обычной формулы (не формулы массива):

```
=СУММ(Данные)/СЧЁТЕСЛИ(Данные;"<>0")
```

Вначале рассчитывается сумма всех значений ячеек в диапазоне, а затем она делится на количество ненулевых ячеек диапазона, полученное с помощью функции СЧЁТЕСЛИ.

Поиск значения в диапазоне

Для того чтобы определить, присутствует ли определенное значение в диапазоне ячеек, необходимо выбрать команду Главная⇒Редактирование⇒Найти и выделить⇒Найти (или нажать комбинацию клавиш <Ctrl+F>), после чего вручную задать параметры поиска. Можно поступить другим образом и решить эту задачу автоматически с помощью формулы массива.

На рис. 15.3 показан рабочий лист со списком имен в диапазоне A3:E22, которому присвоено имя СписокИмен. Формула массива в ячейке D1 ищет имя, введенное в ячейке C1, которой присвоено имя Имя. Если заданное имя присутствует в списке имен, формула выводит текст Найдено. В противном случае она отображает фразу Не найдено.

	A	B	C	D	E
1	Введите имя -->		Dave	Найдено	
2					
3	Al	Daniel	Harold	Lyle	Richard
4	Allen	Dave	Ian	Maggie	Rick
5	Andrew	David	Jack	Margaret	Robert
6	Anthony	Dennis	James	Marilyn	Rod
7	Arthur	Don	Jan	Mark	Roger
8	Barbara	Donald	Jeff	Marvin	Ronald
9	Bernard	Doug	Jeffrey	Mary	Russ
10	Beth	Douglas	Jerry	Matt	Sandra
11	Bill	Ed	Jim	Mel	Scott
12	Bob	Edward	Joe	Merle	Simon
13	Brian	Eric	John	Michael	Stacy
14	Bruce	Fran	Joseph	Michelle	Stephen
15	Cark	Frank	Karl	Mike	Steven
16	Carl	Fred	Kathy	Norman	Stuart
17	Charles	Garv	Keith	Patrick	Susan

Рис. 15.3. Использование формулы массива для определения наличия значения в диапазоне

Формула в ячейке D1 имеет следующий вид:

```
{=ЕСЛИ(ИЛИ(Имя=СписокИмен); "Найдено"; "Не найдено") }
```

Эта формула сравнивает Имя с каждым элементом в диапазоне СписокИмен. Она создает новый массив, который состоит из логических значений ИСТИНА или ЛОЖЬ. Функция ИЛИ возвращает значение ИСТИНА, если хотя бы одно значение в новом массиве — ИСТИНА. Функция ЕСЛИ использует этот результат для определения возвращаемого сообщения.

Более простой вариант этой формулы представлен ниже. Эта формула возвращает значение ИСТИНА, если имя найдено, в противном случае — ЛОЖЬ.

```
{=ИЛИ(Имя=СписокИмен) }
```

Еще один подход предполагает использование функции СЧЁТЕСЛИ:

```
=ЕСЛИ(СЧЁТЕСЛИ(СписокИмен153;Имя)>0;"Найдено";"Не найдено")
```

Подсчет отличающихся значений в двух диапазонах

Представленная ниже формула массива сравнивает соответствующие значения в двух диапазонах (имеющих имена МоиДанные и ТвоиДанные) и возвращает количество различных элементов в этих двух диапазонах. Если содержимое двух диапазонов идентично, формула возвращает нуль.

```
{=СУММ(ЕСЛИ(МоиДанные=ТвоиДанные;0;1) ) }
```

Пример показан на рис. 15.4.



Два диапазона должны иметь одинаковые размер и ориентацию.

Примечание

Данная формула создает новый массив такого же размера, как и сравниваемые диапазоны. Функция ЕСЛИ заполняет этот новый массив нулями и единицами (нуль — если есть различие, единица — если нет различий в соответствующих ячейках). Функция СУММ возвращает сумму значений этого массива.

Рис. 15.4. Использование формулы массива для подсчета количества отличий между двумя диапазонами

```
{=СУММ(1*(МоиДанные<>ТвоиДанные)) }
```

ИСТИНА*1=1

И

ЛОЖЬ * 1 = 0

```
{=МИН (ЕСЛИ (Данные=МАКС (Данные) ; СТРОКА (Данные) ; "" ) ) }
```

Купить книгу на сайте kniga.biz.ua >>>

строку. Функция МИН использует этот массив как второй аргумент и возвращает минимальное значение, которое соответствует номеру строки, в которой находится максимальное значение диапазона Данные.

Если диапазон Данные содержит несколько ячеек с максимальным значением, то возвращается номер первой такой ячейки.

Представленная ниже формула массива подобна предыдущей, но она возвращает действительный адрес ячейки с максимальным значением в диапазоне Данные. В формуле используется функция АДРЕС, которая имеет два аргумента: номер строки и номер столбца.

```
{=АДРЕС(МИН(ЕСЛИ(Данные=МАКС(Данные);СТРОКА(Данные);""));  
СТОЛБЕЦ(Данные))}
```

Если диапазон Данные занимает больше одного столбца, следующая формула массива вернет правильный адрес:

```
{=АДРЕС(МИН(ЕСЛИ(Данные=МАКС(Данные);СТРОКА(Данные);""));  
МИН(ЕСЛИ(Данные=МАКС(Данные);СТОЛБЕЦ(Данные);"")))}
```

Поиск номера строки, в которой находится *n*-е значение, совпадающее с заданным

Приведенная ниже формула ищет в одномерном вертикальном диапазоне Данные значения, которые совпадают с содержимым ячейки Значение, и возвращает номер строки *n*-го совпадающего элемента.

```
{=НАИМЕНЬШИЙ(ЕСЛИ(Данные=Значение;СТРОКА(Данные);"");n)}
```

Функция ЕСЛИ создает новый массив, состоящий из номеров строк, в которых значения из диапазона Данные совпадают со значением Значение. Если значение из диапазона Данные не совпадает со значением Значение, то новый массив содержит пустую текстовую строку. Функция НАИМЕНЬШИЙ обрабатывает этот новый массив и возвращает *n*-й наименьший номер столбца.

Если *n* превышает количество значений в диапазоне Данные, равных Значение, или значение Значение не найдено, формула возвращает ошибку #ЧИСЛО!.

Получение самого длинного текста в диапазоне

Следующая формула отображает содержимое ячейки диапазона Данные, в которой находится наибольшее количество символов. Если самый длинный текст содержится в нескольких ячейках, то возвращается содержимое первой такой ячейки.

```
{=ИНДЕКС(Данные;ПОИСКПОЗ(МАКС(ДЛСТР(Данные));ДЛСТР(Данные);  
ЛОЖЬ);1)}
```

Эта формула работает с двумя массивами, каждый из которых имеет длину соответствующего элемента из диапазона Данные. Функция МАКС находит наибольшее значение, которое соответствует элементу с самым длинным текстом. Функция ПОИСКПОЗ вычисляет смещение ячейки с максимальной длиной. Функция ИНДЕКС возвращает содержимое ячейки, в которой находится текст с наибольшим количеством символов. Эта формула возвращает правильный результат только в том случае, если диапазон Данные состоит из одного столбца (рис. 15.5).

D5	:	X	✓	f _x	{=ИНДЕКС(Данные;ПОИСКПОЗ(
	A	B	C	D	E
1	Самый длинный текст в диапазоне				
2					
3					
4	Январь				
5	Февраль				
6	Март				
7	Апрель				
8	Май				
9	Июнь				
10	Июль				
11	Август				
12	Сентябрь				
13	Октябрь				
14	Ноябрь				
15	Декабрь				

Самый длинный текст: **Сентябрь**

Рис. 15.5. Использование формулы массива для поиска самого длинного текста в диапазоне

Определение допустимых значений диапазона

Перед вами может возникнуть следующая задача: проконтролировать содержимое списка на предмет допустимости в нем значений. Предположим, что список всех допустимых чисел хранится в диапазоне **Основа**. При импорте списка чисел в диапазон **МойСписок** нужно убедиться, что все числа допустимы. Для этого необходимо сравнить элементы в импортируемом списке с элементами в списке **Основа** (рис. 15.6).

F4	:	X	✓	<i>fx</i>	{=ЕНД(ПОИСКПОЗ(ИСТИНА;ЕНД(ПОИСКПОЗ(МойСпис			
	A	B	C	D	E	F	G	
1	Допустимые значения в диапазоне							
2								
3	Основа	МойСписок						
4	AZ-101	AZ-109					Все элементы допустимые?	ЛОЖЬ
5	AZ-102	AZ-105					Недопустимые элементы:	2
6	AZ-103	AZ-109					Первый недопустимый элемент:	AZ-121
7	AZ-104	AZ-107						
8	AZ-105	AZ-121						
9	AZ-106	AZ-122						
10	AZ-107							
11	AZ-108							
12	AZ-109							

Рис. 15.6. Использование формулы массива для подсчета и идентификации элементов, отсутствующих в списке

Приведенная ниже формула массива возвращает значение **ИСТИНА**, если каждый элемент диапазона **МойСписок** найден в диапазоне **Основа**. Оба этих диапазона должны состоять из одного столбца (это не зависит от числа строк):

{=ЕНД(ПОИСКПОЗ(ИСТИНА;ЕНД(ПОИСКПОЗ(МойСписок;Основа;0);0))}

Можно также определить количество недопустимых значений. Иными словами, если требуется подсчитать количество элементов в диапазоне `МойСписок`, которые отсутствуют в диапазоне `Основа`, воспользуйтесь формулой

```
{=СУММ(1*ЕНД(ПОИСКПОЗ(МойСписок;Основа;0)))}
```

Для определения первого недопустимого элемента в диапазоне `МойСписок` используйте формулу

```
{=ИНДЕКС(МойСписок;ПОИСКПОЗ(ИСТИНА;  
ЕНД(ПОИСКПОЗ(МойСписок;Основа;0));0))}
```

Вычисление суммы цифр числа

Я не могу представить себе, зачем это может понадобиться на практике, но это хорошая демонстрация возможностей формул массивов. Для вычисления суммы цифр в положительном целом числе, которое хранится в ячейке `A1`, можно использовать следующую формулу:

```
{=СУММ(ПСТР(A1;СТРОКА(ДВССЫЛ("1:"&ДЛСТР(A1)));1)*1)}
```

Например, если ячейка `A1` содержит значение `409`, приведенная выше формула вернет значение `13` ($4+0+9=13$).

Чтобы понять, как работает эта формула, рассмотрим функцию `СТРОКА` в следующей формуле:

```
{=СТРОКА(ДВССЫЛ("1:"&ДЛСТР(A1)))}
```

Эта формула возвращает массив последовательных целых чисел, который начинается с единицы и заканчивается количеством цифр в числе, находящемся в ячейке `A1`. Например, если ячейка `A1` содержит значение `409`, функция `ДЛСТР` вернет значение `3`, а функцией `СТРОКА` будет сгенерирован следующий массив:

```
{1;2;3}
```



Подробная информация об использовании функции `ДВССЫЛ` приведена в главе 14.

Ссылка

Полученный массив впоследствии используется в качестве второго аргумента функции `ПСТР`. Если в формулу с `ПСТР` подставить полученные в описанном примере значения, она будет упрощена:

```
{ПСТР(409;{1;2;3};1)*1}
```

Эта формула генерирует массив, состоящий из трех элементов:

```
{4;0;9}
```

В результате общая формула с функцией `СУММ` станет более простой и будет иметь следующий вид:

```
=СУММ({4;0;9})
```

Результат вычислений равен `13`.



Примечание

Функция `ПСТР` возвращает текстовую строку, поэтому при преобразовании текста в число элементы массива, находящиеся в аргументе функции `ПСТР`, умножаются на единицу. Альтернативный путь — использование функции `ЗНАЧЕН`, которая также преобразует строку текста, отображающую число, в числовое значение.

	А	В
1	Сумма цифр числа	
2		
3	Число	Сумма цифр
4	132	6
5	9	9
6	111111	6
7	980991	36
8	-980991	36
9	409	13
10		0
11	12	3
12	123	6
13		

Рис. 15.7. Формула массива вычисляет сумму цифр любого целого числа

Заметьте, что описанная выше формула не работает с отрицательными значениями, так как знак “минус” не является цифрой. Ниже приводится формула, в которой исправлен данный недостаток. Для этого используется функция ABS, которая возвращает абсолютное значение числа. На рис. 15.7 представлен рабочий лист, содержащий в ячейке В4 следующую формулу:

```
{=СУММ(ЗНАЧЕН(ПСТР(ABS(A2);СТРОКА(ДВССЫЛ("1:"&ДЛСТР(ABS(A2)))));1)))}
```

Формула в ячейке В4 была скопирована в другие ячейки столбца В с целью вычислить суммы цифр в числах, находящихся в столбце А.

Суммирование округленных значений

На рис. 15.8 показан рабочий лист, который демонстрирует характерную проблему всех электронных таблиц — ошибку округления. Как видно на рисунке, общая сумма в ячейке Е7 не является точной суммой чисел в столбце Е. Значения в столбце Е используют числовой формат, в котором отображаются две цифры после запятой. Фактические значения имеют больше знаков после запятой, и некоторые из них не отображаются. Результирующее влияние погрешностей округления может привести к общему неточному результату. Общее значение составляет \$168,320997, отображаемое — \$168,32.

	А	В	С	Д	Е	Ф	Г	Н
1	Суммирование округленных значений							
2								
3	Описание	Количество	Цена за единицу	Скидка	Сумма			
4	Товар 1	6	\$11,69	5,23%	\$66,47			
5	Товар 2	8	\$9,74	5,23%	\$73,84			
6	Товар 3	3	\$9,85	5,23%	\$28,00			
7	ОБЩАЯ СУММА				\$168,32	<-- неправильная сумма		
8								
9			Сумма округленных значений:		\$168,31			
10								

Рис. 15.8. Использование формулы массива устраняет ошибку округления

Следующая формула массива создает новый массив, который состоит из значений столбца Е, округленных до двух десятичных знаков:

```
{=СУММ(ОКРУГЛ(Е2:Е4;2))}
```

Эта формула возвращает правильную величину суммы округленных значений — \$168,31.

Для устранения таких ошибок округления можно также ввести в ячейки столбца Е уже округленные значения с помощью функции ОКРУГЛ. Этот подход не требует создания формулы массива.

Для суммирования каждого *n*-го элемента диапазона используется следующая формула:

```
{=СУММ(ЕСЛИ(ОСТАТ(СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные156))-1;n)=0;Данные156;""))}
```

Суммирование каждого n -го значения в массиве

Предположим, что существует диапазон значений, в котором необходимо определить сумму каждого третьего значения: первого, четвертого, седьмого и т.д. Один из способов решения этой задачи — внесение адреса ячеек в формулу суммирования. Однако лучшим решением будет использование формулы массива.

Обратимся к данным рис. 15.9. Значения хранятся в диапазоне B4:B22 с именем Данные156, а значение n задается в ячейке D4, названной n .

	A	B	C	D	E
1	Суммирование каждого n -го значения				
2					
3		Данные			
4		1	N=	5	
5		2	Результат=	34	
6		3			
7		4		34	Альтернативный метод
8		5			
9		6			
10		7			
11		8			
12		9			
13		10			
14		11			
15		12			
16		13			
17		14			
18		15			
19		16			
20		17			
21		18			
22		19			

Рис. 15.9. Формула массива возвращает сумму каждого n -го значения в диапазоне

Эта формула создает массив последовательных целых чисел. Функция ОСТАТ принимает его в качестве первого аргумента. Второй аргумент функции ОСТАТ — значение n . Функция ОСТАТ создает другой массив, состоящий из остатков деления каждого последовательного целого числа на n . При равенстве элемента этого массива нулю (это значит, что деление прошло без остатка) происходит суммирование соответствующих значений в диапазоне Данные.

Если принять значение n равным нулю, приведенная выше формула возвращает сообщение об ошибке. Улучшенная формула массива включает функцию ЕСЛИ для успешной работы при $n=0$.

```
{=ЕСЛИ (n=0;0;СУММ (ЕСЛИ (ОСТАТ (СТРОКА (ДВССЫЛ ("1:" &СТРОК (Данные156))) -1;n)=0;Данные156;""))}
```

Эта формула работоспособна только в том случае, когда диапазон Данные состоит из одного столбца значений. Она не работает, если диапазон состоит из более чем одного столбца или даже одной строки.

Приведенная выше формула всегда включает первый элемент диапазона. Например, если $n=5$, будут суммироваться следующие элементы: 1, 6, 11 и 16. Рассмотрим несколько модифицированную версию формулы, которая начинает суммирование именно с n -го элемента. Когда $n=5$, формула суммирует следующие элементы: 5, 10 и 15.

```
{=ЕСЛИ (n=0;0;СУММ (ЕСЛИ (ОСТАТ (СТРОКА (ДВССЫЛ ("1:" &ЧСТРОК (Данные156) ) ) -n; n)=0;Данные156;"") ) ) }
```


Для создания формулы, работающей с горизонтальным диапазоном, необходимо транспонировать массив целых чисел, генерируемых функцией СТРОКА. Модифицированная формула массива, предназначенная для обработки горизонтального диапазона Данные, имеет следующий вид:

```
{=ЕСЛИ (n=0;0;СУММ (ЕСЛИ (ОСТАТ (ТРАНСП (СТРОКА (ДВССЫЛ ("1:" &ЧСТРОК (Данные) ) ) ) -1; n)=0;Данные;"") ) ) }
```



ПРОВЕРКА ПРАВИЛЬНОСТИ ВЫЧИСЛЕНИЯ ФОРМУЛ

Чтобы лучше понять, как выполняется конкретная сложная формула массивов, обратитесь к специальному средству отладки формул. Для этого выделите ячейку с формулой и выберите команду Формулы⇒Зависимости формул⇒Вычислить формулу. На экране откроется диалоговое окно Вычисление формулы, показанное ниже. Щелкните на кнопке Вычислить несколько раз, и вы увидите промежуточные результаты вычисления формулы. Это подобно замедленному, пошаговому воспроизведению процесса вычисления.



Удаление нечисловых символов из текстовой строки

Приведенная ниже формула извлекает числа из текстовой строки. Например, если применить формулу к строке, содержащей текст ABC145Z, будет возвращено значение 145.

```
{=ПСТР (A1; ПОИСКПОЗ (0; (ЕОШИБКА (ПСТР (A1; СТРОКА (ДВССЫЛ ("1:" &ДЛСТР (A1) ) ); 1) *1) *1); 0); ДЛСТР (A1) -СУММ (ЕОШИБКА (ПСТР (A1; СТРОКА (ДВССЫЛ ("1:" &ДЛСТР (A1) ) ); 1) *1) *1) ) }
```

Эта формула работает только с одним внедренным числом. Например, она даст неправильный результат при работе со строкой X45Z99 (вернет 45Z9).

Поиск ближайшего значения в диапазоне

Пусть ячейка с произвольным числом имеет имя Цель157. Для определения ближайшего к нему числа из массива Данные157 воспользуемся формулой массива:

```
{=ИНДЕКС (Данные157; ПОИСКПОЗ (НАИМЕНЬШИЙ (ABS (
Цель157-Данные157) ; 1) ; ABS (Цель157-Данные157) ; 0) ) }
```

Если в массиве существует два числа, ближайших к Цель157, то формула вернет значение первого из них. На рис. 15.10 представлен пример использования этой формулы.

	А	В	С	Д
1	Определение ближайшего значения в диапазоне			
2				
3				
4	-12	Целевое значение:		45
5	-4	Ближайшее значение:		48
6	4			
7	12			
8	20			
9	32			
10	40			
11	48			
12	56			
13	77			

Рис. 15.10. Формула массива возвращает ближайшее значение

В этом примере массив Данные157 находится в диапазоне A4 : A22, значение Цель157 составляет 45. Формула массива находится в ячейке D5 и возвращает ближайшее к значению Цель157 значение, равное 48.

Получение последнего значения в столбце

Предположим, существует рабочий лист, в столбец А которого периодически вносятся новые данные, и нам необходимо отобразить последнее введенное значение. Если столбец А не содержит пустых ячеек, данную задачу можно решить сравнительно просто, не прибегая к использованию формул массива:

```
=СМЕЩ (А1; СЧЁТЗ (А:А) -1; 0)
```

В этой формуле функция СЧЁТЗ используется для подсчета количества непустых ячеек в столбце А. Полученное значение (за вычетом единицы) используется в качестве второго аргумента функции СМЕЩ. Например, пусть последнее значение было введено в строку с номером 100. Функция СЧЁТЗ вернет значение 100, а функция СМЕЩ — значение в ячейке, находящейся на 99 строк ниже ячейки А1.

В то же время столбец А вполне может содержать хотя бы одну пустую ячейку в диапазоне. Это приведет к тому, что предыдущая формула будет выполняться неправильно. Дело в том, что функция СЧЁТЗ не подсчитывает пустые ячейки.

Приведенная ниже формула массива возвращает содержимое последней непустой ячейки столбца А.

```
{=ИНДЕКС (А1:А; МАКС (СТРОКА (А1:А) * (А1:А<>"") ) ) }
```

Естественно, что эту формулу можно модифицировать для работы с любым другим столбцом. Для этого вместо ссылок на столбец A введите ссылки на любой другой столбец.

Если в столбце есть ошибка, эта формула не работает.



Внимание

Эту формулу нельзя размещать в том столбце, который она обрабатывает. Игнорирование этого правила приводит к циклической ссылке. Однако можно немного модифицировать формулу. Например, разместите формулу в первой строке и измените ссылки так, чтобы они начинались со второй строки.



Совет

Существует также альтернативная формула (не формула массива), возвращающая последнее значение в столбце. Приведенная ниже формула возвращает последнюю непустую ячейку столбца A.

=ПРОСМОТР (2; 1/ (A:A<>"") ; A:A)

Единственная проблема состоит в том, что непонятно, как эта формула работает. Но она вполне надежная и, мне кажется, более эффективная, чем формула массива. От формулы массива она отличается тем, что игнорирует ошибочные значения. Фактически она возвращает последнюю непустую ячейку, не содержащую ошибки.

Получение последнего значения в строке

Следующая формула массива аналогична предыдущей, однако возвращает последнюю непустую ячейку строки (в данном случае строки 1):

{=ИНДЕКС (1:1; МАКС (СТОЛБЕЦ (1:1) * (1:1<>""))) }

Для использования этой формулы в другой строке замените ссылку 1:1.

На рис. 15.11 показан пример для последних значений в столбце и строке.

E4	:	X	✓	f _x	{=ИНДЕКС(A1:A501;МАКС(СТРОКА(A1:A501)*(A1:A501<>"")))}				
	A	B	C	D	E	F	G	H	I
1	Поиск последнего значения в строке или столбце								
2									
3					Массив	Не массив			
4	Последняя непустая ячейка в столбце A:				101	101			
5	Последняя непустая ячейка в строке 16:				45	45			
6									
7		5	1						
8		23	2						
9		43	3						
10		2	4						
11			5						
12		2	6						
13		2	7						
14		54	8						
15			9						
16		101	10	45					

Рис. 15.11. Использование формулы массива для возвращения последней непустой ячейки в столбце или строке

Приведенная ниже альтернативная формула (не формула массива) возвращает последнюю непустую и, кроме того, не ошибочную ячейку в строке.

=ПРОСМОТР (2; 1/ (1:1<>"") ; 1:1)

Многоячеечные формулы массивов

В предыдущей главе вы узнали об использовании многоячеечных формул массива, возвращающих несколько значений. В этой главе представлено еще несколько полезных примеров использования таких формул массива. Большинство приведенных формул возвращает несколько или все значения диапазона, реорганизованные в определенном порядке.



На сайте

Все примеры этого раздела содержатся в файлах multicell array formulas.xlsx и Chapter15_RUS.xlsx.

При вводе многоячеечной формулы массива необходимо сначала выделить весь диапазон, а затем ввести формулу и нажать комбинацию клавиш <Ctrl+Shift+Enter>.

Извлечение положительных значений из диапазона

Представленная ниже формула массива работает с одномерным вертикальным массивом, имеющим имя Данные. Формула массива расположена в диапазоне такого же размера, что и диапазон Данные. Для извлечения из диапазона Данные только положительных значений (нулевые и отрицательные значения игнорируются) используется следующая формула массива:

```
{=ИНДЕКС (Данные; НАИМЕНЬШИЙ (ЕСЛИ (Данные>0;
СТРОКА (ДВССЫЛ ("1:" & ЧСТРОК (Данные) ) ) );
СТРОКА (ДВССЫЛ ("1:" & ЧСТРОК (Данные) ) ) ) ) }
```

Как видно на рис. 15.12, эта формула работает, но имеет некоторые недостатки. Диапазон Данные находится в ячейках A5:A24, а формула массива введена в диапазон C5:C24. Однако для ячеек, которые не содержат значения, формула массива возвращает сообщение об ошибке #ЧИСЛО!.

Модифицированная формула массива, введенная в диапазон ячеек E5:E24, использует функцию ЕСЛИОШИБКА во избежание сообщений об ошибках.

```
{=ЕСЛИОШИБКА (ИНДЕКС (Данные; НАИМЕНЬШИЙ (ЕСЛИ (Данные>0;
ДВССЫЛ ( ("1:" & ЧСТРОК (Данные) ) ) );
СТРОКА (ДВССЫЛ ("1:" & ЧСТРОК (Данные) ) ) ) ); "" ) }
```

Извлечение непустых ячеек из диапазона

Формула этого раздела — один из вариантов формулы предыдущего раздела. Она работает с одномерным вертикальным массивом Данные. Формула массива вводится в такой же по размерности диапазон, что и диапазон Данные, и возвращает только непустые ячейки из этого диапазона.

```
{=ЕСЛИОШИБКА (ИНДЕКС (Данные;
НАИМЕНЬШИЙ (ЕСЛИ (Данные<>0;
ДВССЫЛ ( ("1:" & ЧСТРОК (Данные) ) ) );
СТРОКА (ДВССЫЛ ("1:" & ЧСТРОК (Данные) ) ) ) ); "" ) }
```

В более старых версиях программы (до Excel 2007) можно использовать следующую формулу.

```
{=ЕСЛИ(ЕОШ(НАИМЕНЬШИЙ(ЕСЛИ(Данные<>"";
СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные)))));
СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные)))));"";
ИНДЕКС(Данные;НАИМЕНЬШИЙ(ЕСЛИ(Данные<>"";
СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные)))));
СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные))))))}
```

	А	В	С	Д	Е
1	Возвращает только положительные значения диапазона				
2					
3					
4	Данные		Положительные значения		Положительные значения
5	33		33		33
6	-33		44		44
7	44		4		4
8	4		43		43
9	-5		99		99
10	0		5		5
11	43		6		6
12	-1		7		7
13	-2		8		8
14	-3		9		9
15	-33		10		10
16	99		11		11
17	5		12		12
18	6		#ЧИСЛО!		
19	7		#ЧИСЛО!		
20	8		#ЧИСЛО!		
21	9		#ЧИСЛО!		

Рис. 15.12. Использование формулы массива для возвращения из диапазона только положительных значений

Изменение порядка следования элементов в диапазоне на противоположный

Представленная ниже формула массива работает с одномерным вертикальным массивом Данные. Она расположена в диапазоне, имеющем такой же размер, что и диапазон Данные. Для возвращения массива с противоположным порядком следования элементов используется следующая формула.

```
{=ЕСЛИ(ИНДЕКС(Данные;ЧСТРОК(Данные)
СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные))+1))="";"";"";
ИНДЕКС(Данные;ЧСТРОК(Данные)-
СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные))+1)))}
```

Присвоим диапазону A4:A13 имя Данные, а формулу массива введем в диапазон C4:C13. На рис. 15.13 продемонстрирована работа представленной выше формулы массива.

Динамическая сортировка значений диапазона

Предположим, существует рабочий лист с одномерным вертикальным диапазоном Данные. Для сортировки значений массива по убыванию воспользуемся следующей формулой массива. Следует заметить, что данная формула работает только с числовыми значениями, но не с текстом:

```
{=НАИБОЛЬШИЙ(Данные;СТРОКА(ДВССЫЛ("1:"&ЧСТРОК(Данные))))}
```

	А	В	С
1	Изменение порядка элементов в диапазоне		
2			
3	Ввод данных		Обращение
4	один		десять
5	два		девять
6	три		восемь
7	четыре		семь
8	пять		шесть
9	шесть		пять
10	семь		четыре
11	восемь		три
12	девять		два
13	десять		один
14			

Рис. 15.13. Порядок следования элементов массива изменяется на противоположный с помощью формулы массива

На рис. 15.14 показан пример; диапазон ввода находится в столбце А; формула массива введена в столбец С.

Эта формула может оказаться полезной, если требуется отсортировать данные немедленно. Начните с присвоения диапазону данных имени Данные, а затем введите формулу в другой диапазон данных, размещенный в тех же строках.

Если в исходной ячейке не содержится данных, формула массива отображает значение ошибки #ЧИСЛО!. В следующей, модифицированной формуле используется функция ЕСЛИОШИБКА. Она позволяет избежать отображения на экране сообщения об ошибке. Эта формула используется в столбце Е на рис. 15.11.

```
{=ЕСЛИОШИБКА (НАИБОЛЬШИЙ (Данные ;  
СТРОКА (ДВССЫЛ ("1:"&ЧСТРОК (Данные) ) ) ) ; "" ) }
```

Эта формула должна вводиться в такой же по размеру диапазон, что и диапазон Данные. Она тоже сортирует только числовые значения.

Для сортировки значений Данные по возрастанию можно использовать такую формулу:

```
{=НАИМЕНЬШИЙ (Данные ; СТРОКА (ДВССЫЛ ("1:"&ЧСТРОК (Данные) ) ) ) }
```

Возвращение списка уникальных значений диапазона

Если существует одномерный диапазон Данные, то для вывода списка уникальных элементов этого диапазона можно воспользоваться следующей формулой.

```
{=ИНДЕКС (Данные ; НАИМЕНЬШИЙ (ЕСЛИ (ПОИСКПОЗ (Данные ;  
Данные ; 0)=СТРОКА (ДВССЫЛ ("1:"&ЧСТРОК (Данные) ) ) ;  
ПОИСКПОЗ (Данные ; Данные ; 0) ; "" ) ;  
СТРОКА (ДВССЫЛ ("1:"&ЧСТРОК (Данные) ) ) ) }
```

Эта формула не работает, если диапазон Данные содержит хотя бы одну пустую ячейку. В незаполненных ячейках формулы массива будет отображаться сообщение об ошибке #ЧИСЛО!. На рис. 15.15 представлен пример применения формулы массива. Диапазон А5:А23 имеет имя Данные, а формула массива введена в диапазон С5:С23. Обратите внимание на то, что в незаполненных ячейках отображается ошибка #ЧИСЛО!.

	А	В	С	Д	Е
1	Динамическое изменение порядка элементов в диапазоне				
2					
3					
4	Ввод данных		Сортировка (с ошибками)		Сортировка (без ошибок)
5	44		233		233
6	25		105		105
7	89		89		89
8	43		55		55
9	31		44		44
10	105		43		43
11			31		31
12	55		25		25
13	9		9		9
14	233		#ЧИСЛО!		
15			#ЧИСЛО!		
16			#ЧИСЛО!		
17			#ЧИСЛО!		
18					

Рис. 15.14. Формула массива, введенная в столбце С, сортирует значения, введенные в столбец А

Чтобы избежать отображения значения ошибки, лучше воспользоваться функцией ЕСЛИОШИБКА.

```
{ ЕСЛИОШИБКА (ИНДЕКС (Данные; НАИМЕНЬШИЙ (ЕСЛИ
(ПОИСКПОЗ (Данные; Данные; 0) =СТРОКА (ДВССЫЛ
(""&ЧСТРОК (Данные) ) ) ;
ПОИСКПОЗ (Данные; Данные; 0) ; "" ) ) ) ; "" ) }
```

На рис. 15.15 эта формула используется в столбце Е.

	А	В	С	Д	Е
4	Ввод данных		Уникальные элементы		Уникальные элементы
5	Собака		Собака		Собака
6	Собака		Мартышка		Мартышка
7	Собака		Кошка		Кошка
8	Собака		Слон		Слон
9	Собака		Попугай		Попугай
10	Собака		#ЧИСЛО!		
11	Собака		#ЧИСЛО!		
12	Собака		#ЧИСЛО!		
13	Собака		#ЧИСЛО!		
14	Мартышка		#ЧИСЛО!		
15	Кошка		#ЧИСЛО!		
16	Слон		#ЧИСЛО!		
17	Слон		#ЧИСЛО!		
18	Слон		#ЧИСЛО!		
19	Попугай		#ЧИСЛО!		
20	Попугай		#ЧИСЛО!		
21	Попугай		#ЧИСЛО!		
22	Собака		#ЧИСЛО!		
23	Мартышка		#ЧИСЛО!		
24					

Рис. 15.15. Формула массива применяется для создания списка уникальных элементов диапазона данных

Отображение календаря в диапазоне

На рис. 15.16 представлен результат одной из формул, отображающей “живой” календарь в диапазоне ячеек. Если изменить дату верхней части диапазона, календарь автоматически отобразит даты соответствующего месяца и года.



На сайте

Рабочая книга данного примера находится в файле `array formula calendar.xlsx`.

D6		=ЕСЛИ(МЕСЯЦ(ДАТА(ГОД(В2);МЕСЯЦ(В2);1))<>МЕСЯЦ(ДАТА(ГОД(В2);МЕСЯЦ(В2);1)-(ДЕНЬНЕД(ДАТА(ГОД(В2);МЕСЯЦ(В2);1))-1)+{0;1;2;3;4;5}*7+{1;2;3;4;5;6;7}-1);"";ДАТА(ГОД(В2);МЕСЯЦ(В2);1)-(ДЕНЬНЕД(ДАТА(ГОД(В2);МЕСЯЦ(В2);1))-1)+{0;1;2;3;4;5}*7+{1;2;3;4;5;6;7}-1))												
	A	B	C	D	E	F	G	H	I	J	K	L	M	
1														
2		Сентябрь 2010												
3		Вск	Пнд	Втр	Срд	Чтв	Птн	Сбт						
4					1	2	3	4						
5		5	6	7	8	9	10	11						
6		12	13	14	15	16	17	18						
7		19	20	21	22	23	24	25						
8		26	27	28	29	30								
9														

Рис. 15.16. Отображение календаря с помощью одной формулы массива

1. Отформатируйте диапазон В4 : Н9 с пользовательским форматом d. В таком формате на экране будут отображаться только даты. Для выбора этого формата воспользуйтесь категорией **Дополнительно** диалогового окна форматирования ячеек.
2. Скорректируйте ширину столбцов согласно своим предпочтениям.

Изменение месяца и года в ячейке В2 приводит к автоматическому обновлению календаря. После создания календаря можно скопировать его диапазон в любое место рабочего листа или книги.

Формула массива фактически возвращает даты, но ячейки отформатированы так, чтобы отображать только дни. Также обратите внимание на то, что в формулах массива используются массивы констант.



Ссылка

Массивы констант рассматриваются в главе 14.

Рассмотрим еще одну версию приведенной выше формулы. Функция ЕСЛИ в исходной формуле проверяет каждую дату на принадлежность текущему месяцу. Если это не так, функция ЕСЛИ возвращает пустую строку.

Приведенную выше формулу массива можно несколько упростить, удалив функцию ЕСЛИ.

=ДАТА(ГОД(В2);МЕСЯЦ(В2);1)-(ДЕНЬНЕД(ДАТА(ГОД(В2);МЕСЯЦ(В2);1))-1)+{0;1;2;3;4;5}*7+{1;2;3;4;5;6;7}-1

Эта версия формулы отображает также дни предыдущего и следующего месяцев.

На рис. 15.17 показаны двенадцать экземпляров календаря на основе формул массивов для всего года.

V20 : X ✓ f {=ЕСЛИ(МЕСЯЦ(ДАТА(ГОД(\$I4);МЕСЯЦ(\$I4);1))<>МЕСЯЦ(ДАТА(ГОД(\$I4);МЕСЯЦ(\$I4);1))-1)}

2013						
Январь						
Вс	Пн	Вт	Ср	Чт	Пт	Сб
		1	2	3	4	5
6	7	8	9	10	11	12
13	14	15	16	17	18	19
20	21	22	23	24	25	26
27	28	29	30	31		
Февраль						
Вс	Пн	Вт	Ср	Чт	Пт	Сб
					1	2
3	4	5	6	7	8	9
10	11	12	13	14	15	16
17	18	19	20	21	22	23
24	25	26	27	28		
Март						
Вс	Пн	Вт	Ср	Чт	Пт	Сб
					1	2
3	4	5	6	7	8	9
10	11	12	13	14	15	16
17	18	19	20	21	22	23
24	25	26	27	28	29	30
31						
Апрель						
Вс	Пн	Вт	Ср	Чт	Пт	Сб
	1	2	3	4	5	6
7	8	9	10	11	12	13
14	15	16	17	18	19	20
21	22	23	24	25	26	27
28	29	30				
Май						
Вс	Пн	Вт	Ср	Чт	Пт	Сб
			1	2	3	4
5	6	7	8	9	10	11
12	13	14	15	16	17	18
19	20	21	22	23	24	25
26	27	28	29	30	31	
Июнь						
Вс	Пн	Вт	Ср	Чт	Пт	Сб
						1
2	3	4	5	6	7	8
9	10	11	12	13	14	15
16	17	18	19	20	21	22
23	24	25	26	27	28	29
30						
Июль						
Вс	Пн	Вт	Ср	Чт	Пт	Сб
	1	2	3	4	5	6
Август						
Вс	Пн	Вт	Ср	Чт	Пт	Сб
				1	2	3
Сентябрь						
Вс	Пн	Вт	Ср	Чт	Пт	Сб
1	2	3	4	5	6	7

Лист1 За год (+)

Рис. 15.17. Календарь на год, состоящий из формул массивов